

# NAG Library Routine Document

## D03PEF

**Note:** before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

### 1 Purpose

D03PEF integrates a system of linear or nonlinear, first-order, time-dependent partial differential equations (PDEs) in one space variable. The spatial discretization is performed using the Keller box scheme and the method of lines is employed to reduce the PDEs to a system of ordinary differential equations (ODEs). The resulting system is solved using a Backward Differentiation Formula (BDF) method.

### 2 Specification

```

SUBROUTINE D03PEF(NPDE, TS, TOUT, PDEDEF, BNDARY, U, NPTS, X, NLEFT,
1          ACC, RSAVE, LRSAVE, ISAVE, LISAVE, ITASK, ITRACE, IND,
2          IFAIL)

  INTEGER      NPDE, NPTS, NLEFT, LRSAVE, ISAVE(LISAVE), LISAVE,
1          ITASK, ITRACE, IND, IFAIL
  double precision TS, TOUT, U(NPDE,NPTS), X(NPTS), ACC, RSAVE(LRSAVE)
  EXTERNAL    PDEDEF, BNDARY

```

### 3 Description

D03PEF integrates the system of first-order PDEs

$$G_i(x, t, U, U_x, U_t) = 0, \quad i = 1, 2, \dots, \text{NPDE}. \quad (1)$$

In particular the functions  $G_i$  must have the general form

$$G_i = \sum_{j=1}^{\text{NPDE}} P_{i,j} \frac{\partial U_j}{\partial t} + Q_i, \quad i = 1, 2, \dots, \text{NPDE}, \quad a \leq x \leq b, t \geq t_0, \quad (2)$$

where  $P_{i,j}$  and  $Q_i$  depend on  $x, t, U, U_x$  and the vector  $U$  is the set of solution values

$$U(x, t) = [U_1(x, t), \dots, U_{\text{NPDE}}(x, t)]^T, \quad (3)$$

and the vector  $U_x$  is its partial derivative with respect to  $x$ . Note that  $P_{i,j}$  and  $Q_i$  must not depend on  $\frac{\partial U}{\partial t}$ .

The integration in time is from  $t_0$  to  $t_{\text{out}}$ , over the space interval  $a \leq x \leq b$ , where  $a = x_1$  and  $b = x_{\text{NPTS}}$  are the leftmost and rightmost points of a user-defined mesh  $x_1, x_2, \dots, x_{\text{NPTS}}$ . The mesh should be chosen in accordance with the expected behaviour of the solution.

The PDE system which is defined by the functions  $G_i$  must be specified in PDEDEF.

The initial values of the functions  $U(x, t)$  must be given at  $t = t_0$ . For a first-order system of PDEs, only one boundary condition is required for each PDE component  $U_i$ . The NPDE boundary conditions are separated into  $n_a$  at the left-hand boundary  $x = a$ , and  $n_b$  at the right-hand boundary  $x = b$ , such that  $n_a + n_b = \text{NPDE}$ . The position of the boundary condition for each component should be chosen with care; the general rule is that if the characteristic direction of  $U_i$  at the left-hand boundary (say) points into the interior of the solution domain, then the boundary condition for  $U_i$  should be specified at the left-hand boundary. Incorrect positioning of boundary conditions generally results in initialization or integration difficulties in the underlying time integration routines.

The boundary conditions have the form:

$$G_i^L(x, t, U, U_t) = 0 \quad \text{at } x = a, \quad i = 1, 2, \dots, n_a \quad (4)$$

at the left-hand boundary, and

$$G_i^R(x, t, U, U_t) = 0 \quad \text{at } x = b, \quad i = 1, 2, \dots, n_b \quad (5)$$

at the right-hand boundary.

Note that the functions  $G_i^L$  and  $G_i^R$  must not depend on  $U_x$ , since spatial derivatives are not determined explicitly in the Keller box scheme (see Keller (1970)). If the problem involves derivative (Neumann) boundary conditions then it is generally possible to restate such boundary conditions in terms of permissible variables. Also note that  $G_i^L$  and  $G_i^R$  must be linear with respect to time derivatives, so that the boundary conditions have the general form

$$\sum_{j=1}^{\text{NPDE}} E_{i,j}^L \frac{\partial U_j}{\partial t} + S_i^L = 0, \quad i = 1, 2, \dots, n_a \quad (6)$$

at the left-hand boundary, and

$$\sum_{j=1}^{\text{NPDE}} E_{i,j}^R \frac{\partial U_j}{\partial t} + S_i^R = 0, \quad i = 1, 2, \dots, n_b \quad (7)$$

at the right-hand boundary, where  $E_{i,j}^L$ ,  $E_{i,j}^R$ ,  $S_i^L$ , and  $S_i^R$  depend on  $x$ ,  $t$  and  $U$  only.

The boundary conditions must be specified in BNDARY.

The problem is subject to the following restrictions:

- (i)  $t_0 < t_{\text{out}}$ , so that integration is in the forward direction;
- (ii)  $P_{i,j}$  and  $Q_i$  must not depend on any time derivatives;
- (iii) The evaluation of the function  $G_i$  is done at the mid-points of the mesh intervals by calling the PDEDEF for each mid-point in turn. Any discontinuities in the function **must** therefore be at one or more of the mesh points  $x_1, x_2, \dots, x_{\text{NPTS}}$ ;
- (iv) At least one of the functions  $P_{i,j}$  must be nonzero so that there is a time derivative present in the problem.

In this method of lines approach the Keller box scheme (see Keller (1970)) is applied to each PDE in the space variable only, resulting in a system of ODEs in time for the values of  $U_i$  at each mesh point. In total there are  $\text{NPDE} \times \text{NPTS}$  ODEs in the time direction. This system is then integrated forwards in time using a BDF method.

## 4 References

- Berzins M (1990) Developments in the NAG Library software for parabolic equations *Scientific Software Systems* (eds J C Mason and M G Cox) 59–72 Chapman and Hall
- Berzins M, Dew P M and Furzeland R M (1989) Developing software for time-dependent problems using the method of lines and differential-algebraic integrators *Appl. Numer. Math.* **5** 375–397
- Keller H B (1970) A new difference scheme for parabolic problems *Numerical Solutions of Partial Differential Equations* (ed J Bramble) **2** 327–350 Academic Press
- Pennington S V and Berzins M (1994) New NAG Library software for first-order partial differential equations *ACM Trans. Math. Softw.* **20** 63–99

## 5 Parameters

- 1: NPDE – INTEGER *Input*  
*On entry:* the number of PDEs in the system to be solved.  
*Constraint:*  $\text{NPDE} \geq 1$ .

- 2: TS – **double precision** Input/Output  
*On entry:* the initial value of the independent variable  $t$ .  
*Constraint:* TS < TOUT.  
*On exit:* the value of  $t$  corresponding to the solution values in U. Normally TS = TOUT.
- 3: TOUT – **double precision** Input  
*On entry:* the final value of  $t$  to which the integration is to be carried out.
- 4: PDEDEF – SUBROUTINE, supplied by the user. External Procedure  
PDEDEF must compute the functions  $G_i$  which define the system of PDEs. PDEDEF is called approximately midway between each pair of mesh points in turn by D03PEF.

The specification of PDEDEF is:

```
SUBROUTINE PDEDEF(NPDE, T, X, U, UT, UX, RES, IRES)
  INTEGER          NPDE, IRES
  double precision T, X, U(NPDE), UT(NPDE), UX(NPDE), RES(NPDE)
```

- 1: NPDE – INTEGER Input  
*On entry:* the number of PDEs in the system.
- 2: T – **double precision** Input  
*On entry:* the current value of the independent variable  $t$ .
- 3: X – **double precision** Input  
*On entry:* the current value of the space variable  $x$ .
- 4: U(NPDE) – **double precision** array Input  
*On entry:* U( $i$ ) contains the value of the component  $U_i(x, t)$ , for  $i = 1, 2, \dots, \text{NPDE}$ .
- 5: UT(NPDE) – **double precision** array Input  
*On entry:* UT( $i$ ) contains the value of the component  $\frac{\partial U_i(x, t)}{\partial t}$ , for  $i = 1, 2, \dots, \text{NPDE}$ .
- 6: UX(NPDE) – **double precision** array Input  
*On entry:* UX( $i$ ) contains the value of the component  $\frac{\partial U_i(x, t)}{\partial x}$ , for  $i = 1, 2, \dots, \text{NPDE}$ .
- 7: RES(NPDE) – **double precision** array Output  
*On exit:* RES( $i$ ) must contain the  $i$ th component of  $G$ , for  $i = 1, 2, \dots, \text{NPDE}$ , where  $G$  is defined as

$$G_i = \sum_{j=1}^{\text{NPDE}} P_{i,j} \frac{\partial U_j}{\partial t}, \quad (8)$$

i.e., only terms depending explicitly on time derivatives, or

$$G_i = \sum_{j=1}^{\text{NPDE}} P_{i,j} \frac{\partial U_j}{\partial t} + Q_i, \quad (9)$$

i.e., all terms in equation (2).

The definition of  $G$  is determined by the input value of IRES.

|    |                                                                                                                                                                                                                                                                                                                                                               |                     |
|----|---------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------|---------------------|
| 8: | IRES – INTEGER                                                                                                                                                                                                                                                                                                                                                | <i>Input/Output</i> |
|    | <i>On entry:</i> the form of $G_i$ that must be returned in the array RES.                                                                                                                                                                                                                                                                                    |                     |
|    | IRES = -1<br>Equation (8) must be used.                                                                                                                                                                                                                                                                                                                       |                     |
|    | IRES = 1<br>Equation (9) must be used.                                                                                                                                                                                                                                                                                                                        |                     |
|    | <i>On exit:</i> should usually remain unchanged. However, you may set IRES to force the integration routine to take certain actions, as described below:                                                                                                                                                                                                      |                     |
|    | IRES = 2<br>Indicates to the integrator that control should be passed back immediately to the calling subroutine with the error indicator set to IFAIL = 6.                                                                                                                                                                                                   |                     |
|    | IRES = 3<br>Indicates to the integrator that the current time step should be abandoned and a smaller time step used instead. You may wish to set IRES = 3 when a physically meaningless input or output value has been generated. If you consecutively set IRES = 3, then D03PEF returns to the calling subroutine with the error indicator set to IFAIL = 4. |                     |

PDEDEF must be declared as EXTERNAL in the (sub)program from which D03PEF is called. Parameters denoted as *Input* must **not** be changed by this procedure.

- 5: BNDARY – SUBROUTINE, supplied by the user. *External Procedure*

BNDARY must compute the functions  $G_i^L$  and  $G_i^R$  which define the boundary conditions as in equations (4) and (5).

The specification of BNDARY is:

```
SUBROUTINE BNDARY(NPDE, T, IBND, NOBC, U, UT, RES, IRES)
```

```
INTEGER NPDE, IBND, NOBC, IRES
```

```
double precision T, U(NPDE), UT(NPDE), RES(NOBC)
```

|    |                                                                                                                                                   |              |
|----|---------------------------------------------------------------------------------------------------------------------------------------------------|--------------|
| 1: | NPDE – INTEGER                                                                                                                                    | <i>Input</i> |
|    | <i>On entry:</i> the number of PDEs in the system.                                                                                                |              |
| 2: | T – <b>double precision</b>                                                                                                                       | <i>Input</i> |
|    | <i>On entry:</i> the current value of the independent variable $t$ .                                                                              |              |
| 3: | IBND – INTEGER                                                                                                                                    | <i>Input</i> |
|    | <i>On entry:</i> determines the position of the boundary conditions.                                                                              |              |
|    | IBND = 0<br>BNDARY must compute the left-hand boundary condition at $x = a$ .                                                                     |              |
|    | IBND $\neq$ 0<br>Indicates that BNDARY must compute the right-hand boundary condition at $x = b$ .                                                |              |
| 4: | NOBC – INTEGER                                                                                                                                    | <i>Input</i> |
|    | <i>On entry:</i> specifies the number of boundary conditions at the boundary specified by IBND.                                                   |              |
| 5: | U(NPDE) – <b>double precision</b> array                                                                                                           | <i>Input</i> |
|    | <i>On entry:</i> U( $i$ ) contains the value of the component $U_i(x, t)$ at the boundary specified by IBND, for $i = 1, 2, \dots, \text{NPDE}$ . |              |

|    |                                                                                                                                                                                                                                                                                                                                                               |                     |
|----|---------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------|---------------------|
| 6: | UT(NPDE) – <b>double precision</b> array                                                                                                                                                                                                                                                                                                                      | <i>Input</i>        |
|    | <i>On entry:</i> UT( <i>i</i> ) contains the value of the component $\frac{\partial U_i(x,t)}{\partial t}$ at the boundary specified by IBND, for $i = 1, 2, \dots, \text{NPDE}$ .                                                                                                                                                                            |                     |
| 7: | RES(NOBC) – <b>double precision</b> array                                                                                                                                                                                                                                                                                                                     | <i>Output</i>       |
|    | <i>On exit:</i> RES( <i>i</i> ) must contain the <i>i</i> th component of $G^L$ or $G^R$ , depending on the value of IBND, for $i = 1, 2, \dots, \text{NOBC}$ , where $G^L$ is defined as                                                                                                                                                                     |                     |
|    | $G_i^L = \sum_{j=1}^{\text{NPDE}} E_{i,j}^L \frac{\partial U_j}{\partial t}, \quad (10)$                                                                                                                                                                                                                                                                      |                     |
|    | i.e., only terms depending explicitly on time derivatives, or                                                                                                                                                                                                                                                                                                 |                     |
|    | $G_i^L = \sum_{j=1}^{\text{NPDE}} E_{i,j}^L \frac{\partial U_j}{\partial t} + S_i^L, \quad (11)$                                                                                                                                                                                                                                                              |                     |
|    | i.e., all terms in equation (6), and similarly for $G_i^R$ .<br>The definitions of $G^L$ and $G^R$ are determined by the input value of IRES.                                                                                                                                                                                                                 |                     |
| 8: | IRES – INTEGER                                                                                                                                                                                                                                                                                                                                                | <i>Input/Output</i> |
|    | <i>On entry:</i> the form $G_i^L$ (or $G_i^R$ ) that must be returned in the array RES.                                                                                                                                                                                                                                                                       |                     |
|    | IRES = -1<br>Equation (10) must be used.                                                                                                                                                                                                                                                                                                                      |                     |
|    | IRES = 1<br>Equation (11) must be used.                                                                                                                                                                                                                                                                                                                       |                     |
|    | <i>On exit:</i> should usually remain unchanged. However, you may set IRES to force the integration routine to take certain actions, as described below:                                                                                                                                                                                                      |                     |
|    | IRES = 2<br>Indicates to the integrator that control should be passed back immediately to the calling subroutine with the error indicator set to IFAIL = 6.                                                                                                                                                                                                   |                     |
|    | IRES = 3<br>Indicates to the integrator that the current time step should be abandoned and a smaller time step used instead. You may wish to set IRES = 3 when a physically meaningless input or output value has been generated. If you consecutively set IRES = 3, then D03PEF returns to the calling subroutine with the error indicator set to IFAIL = 4. |                     |

BNDARY must be declared as EXTERNAL in the (sub)program from which D03PEF is called. Parameters denoted as *Input* must **not** be changed by this procedure.

- |    |                                                                                                                                      |                     |
|----|--------------------------------------------------------------------------------------------------------------------------------------|---------------------|
| 6: | U(NPDE,NPTS) – <b>double precision</b> array                                                                                         | <i>Input/Output</i> |
|    | <i>On entry:</i> the initial values of $U(x,t)$ at $t = \text{TS}$ and the mesh points $X(j)$ , for $j = 1, 2, \dots, \text{NPTS}$ . |                     |
|    | <i>On exit:</i> $U(i,j)$ will contain the computed solution at $t = \text{TS}$ .                                                     |                     |
| 7: | NPTS – INTEGER                                                                                                                       | <i>Input</i>        |
|    | <i>On entry:</i> the number of mesh points in the interval $[a,b]$ .                                                                 |                     |
|    | <i>Constraint:</i> $\text{NPTS} \geq 3$ .                                                                                            |                     |

- 8: X(NPTS) – **double precision** array *Input*  
*On entry:* the mesh points in the spatial direction. X(1) must specify the left-hand boundary,  $a$ , and X(NPTS) must specify the right-hand boundary,  $b$ .  
*Constraint:*  $X(1) < X(2) < \dots < X(NPTS)$ .
- 9: NLEFT – INTEGER *Input*  
*On entry:* the number  $n_a$  of boundary conditions at the left-hand mesh point X(1).  
*Constraint:*  $0 \leq NLEFT \leq NPDE$ .
- 10: ACC – **double precision** *Input*  
*On entry:* a positive quantity for controlling the local error estimate in the time integration. If  $E(i, j)$  is the estimated error for  $U_i$  at the  $j$ th mesh point, the error test is:  

$$|E(i, j)| = ACC \times (1.0 + |U(i, j)|).$$
*Constraint:*  $ACC > 0.0$ .
- 11: RSAVE(LRSAVE) – **double precision** array *Communication Array*  
If  $IND = 0$ , RSAVE need not be set on entry.  
If  $IND = 1$ , RSAVE must be unchanged from the previous call to the routine because it contains required information about the iteration.
- 12: LRSAVE – INTEGER *Input*  
*On entry:* the dimension of the array RSAVE as declared in the (sub)program from which D03PEF is called.  
*Constraint:*  
 $LRSAVE \geq (4 \times NPDE + NLEFT + 14) \times NPDE \times NPTS + (3 \times NPDE + 21) \times NPDE + 7 \times NPTS + 54$ .
- 13: ISAVE(LISAVE) – INTEGER array *Communication Array*  
If  $IND = 0$ , ISAVE need not be set on entry.  
If  $IND = 1$ , ISAVE must be unchanged from the previous call to the routine because it contains required information about the iteration. In particular:  
ISAVE(1)  
Contains the number of steps taken in time.  
ISAVE(2)  
Contains the number of residual evaluations of the resulting ODE system used. One such evaluation involves computing the PDE functions at all the mesh points, as well as one evaluation of the functions in the boundary conditions.  
ISAVE(3)  
Contains the number of Jacobian evaluations performed by the time integrator.  
ISAVE(4)  
Contains the order of the last backward differentiation formula method used.  
ISAVE(5)  
Contains the number of Newton iterations performed by the time integrator. Each iteration involves an ODE residual evaluation followed by a back-substitution using the *LU* decomposition of the Jacobian matrix.

- 14: LISAVE – INTEGER *Input*  
*On entry:* the dimension of the array ISAVE as declared in the (sub)program from which D03PEF is called.  
*Constraint:*  $\text{LISAVE} \geq \text{NPDE} \times \text{NPTS} + 24$ .
- 15: ITASK – INTEGER *Input*  
*On entry:* specifies the task to be performed by the ODE integrator.  
 $\text{ITASK} = 1$   
 Normal computation of output values  $U$  at  $t = \text{TOUT}$ .  
 $\text{ITASK} = 2$   
 Take one step and return.  
 $\text{ITASK} = 3$   
 Stop at the first internal integration point at or beyond  $t = \text{TOUT}$ .  
*Constraint:*  $\text{ITASK} = 1, 2 \text{ or } 3$ .
- 16: ITRACE – INTEGER *Input*  
*On entry:* the level of trace information required from D03PEF and the underlying ODE solver as follows:  
 $\text{ITRACE} \leq -1$   
 No output is generated.  
 $\text{ITRACE} = 0$   
 Only warning messages from the PDE solver are printed on the current error message unit (see X04AAF).  
 $\text{ITRACE} = 1$   
 Output from the underlying ODE solver is printed on the current advisory message unit (see X04ABF). This output contains details of Jacobian entries, the nonlinear iteration and the time integration during the computation of the ODE system.  
 $\text{ITRACE} = 2$   
 Output from the underlying ODE solver is similar to that produced when  $\text{ITRACE} = 1$ , except that the advisory messages are given in greater detail.  
 $\text{ITRACE} \geq 3$   
 Output from the underlying ODE solver is similar to that produced when  $\text{ITRACE} = 2$ , except that the advisory messages are given in greater detail.  
 You are advised to set  $\text{ITRACE} = 0$ , unless you are experienced with sub-chapter D02M–N.
- 17: IND – INTEGER *Input/Output*  
*On entry:* indicates whether this is a continuation call or a new integration.  
 $\text{IND} = 0$   
 Starts or restarts the integration in time.  
 $\text{IND} = 1$   
 Continues the integration after an earlier exit from the routine. In this case, only the parameters TOUT and IFAIL should be reset between calls to D03PEF.  
*Constraint:*  $\text{IND} = 0 \text{ or } 1$ .  
*On exit:*  $\text{IND} = 1$ .
- 18: IFAIL – INTEGER *Input/Output*  
*On entry:* IFAIL must be set to 0, -1 or 1. If you are unfamiliar with this parameter you should refer to Section 3.3 in the Essential Introduction for details.

*On exit:* IFAIL = 0 unless the routine detects an error (see Section 6).

For environments where it might be inappropriate to halt program execution when an error is detected, the value  $-1$  or  $1$  is recommended. If the output of error messages is undesirable, then the value  $1$  is recommended. Otherwise, if you are not familiar with this parameter, the recommended value is  $0$ . **When the value  $-1$  or  $1$  is used it is essential to test the value of IFAIL on exit.**

## 6 Error Indicators and Warnings

If on entry IFAIL = 0 or  $-1$ , explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

IFAIL = 1

On entry, TOUT  $\leq$  TS,  
 or (TOUT – TS) is too small,  
 or ITASK  $\neq$  1, 2 or 3,  
 or mesh points  $X(i)$  are not ordered correctly,  
 or NPTS  $< 3$ ,  
 or NPDE  $< 1$ ,  
 or NLEFT is not in the range 0 to NPDE,  
 or ACC  $\leq 0.0$ ,  
 or IND  $\neq 0$  or 1,  
 or LRSAVE is too small,  
 or LISAVE is too small,  
 or D03PEF called initially with IND = 1.

IFAIL = 2

The underlying ODE solver cannot make any further progress, across the integration range from the current point  $t = TS$  with the supplied value of ACC. The components of U contain the computed values at the current point  $t = TS$ .

IFAIL = 3

In the underlying ODE solver, there were repeated errors or corrector convergence test failures on an attempted step, before completing the requested task. The problem may have a singularity or ACC is too small for the integration to continue. Incorrect positioning of boundary conditions may also result in this error. Integration was successful as far as  $t = TS$ .

IFAIL = 4

In setting up the ODE system, the internal initialization routine was unable to initialize the derivative of the ODE system. This could be due to the fact that IRES was repeatedly set to 3 in the PDEDEF or BNDARY, when the residual in the underlying ODE solver was being evaluated. Incorrect positioning of boundary conditions may also result in this error.

IFAIL = 5

In solving the ODE system, a singular Jacobian has been encountered. You should check their problem formulation.

IFAIL = 6

When evaluating the residual in solving the ODE system, IRES was set to 2 in one of PDEDEF or BNDARY. Integration was successful as far as  $t = TS$ .

IFAIL = 7

The value of ACC is so small that the routine is unable to start the integration in time.

IFAIL = 8

In either, PDEDEF or BNDARY, IRES was set to an invalid value.

IFAIL = 9 (D02NNF)

A serious error has occurred in an internal call to the specified routine. Check the problem specification and all parameters and array dimensions. Setting ITRACE = 1 may provide more information. If the problem persists, contact NAG.

IFAIL = 10

The required task has been completed, but it is estimated that a small change in ACC is unlikely to produce any change in the computed solution. (Only applies when you are not operating in one step mode, that is when ITASK  $\neq$  2.)

IFAIL = 11

An error occurred during Jacobian formulation of the ODE system (a more detailed error description may be directed to the current advisory message unit).

## 7 Accuracy

D03PEF controls the accuracy of the integration in the time direction but not the accuracy of the approximation in space. The spatial accuracy depends on both the number of mesh points and on their distribution in space. In the time integration only the local error over a single step is controlled and so the accuracy over a number of steps cannot be guaranteed. You should therefore test the effect of varying the accuracy parameter, ACC.

## 8 Further Comments

The Keller box scheme can be used to solve higher-order problems which have been reduced to first-order by the introduction of new variables (see the example problem in D03PKF). In general, a second-order problem can be solved with slightly greater accuracy using the Keller box scheme instead of a finite-difference scheme (D03PCF/D03PCA or D03PHF/D03PHA for example), but at the expense of increased CPU time due to the larger number of function evaluations required.

It should be noted that the Keller box scheme, in common with other central-difference schemes, may be unsuitable for some hyperbolic first-order problems such as the apparently simple linear advection equation  $U_t + aU_x = 0$ , where  $a$  is a constant, resulting in spurious oscillations due to the lack of dissipation. This type of problem requires a discretization scheme with upwind weighting (D03PFF for example), or the addition of a second-order artificial dissipation term.

The time taken depends on the complexity of the system and on the accuracy requested.

## 9 Example

This example is the simple first-order system

$$\frac{\partial U_1}{\partial t} + \frac{\partial U_1}{\partial x} + \frac{\partial U_2}{\partial x} = 0,$$

$$\frac{\partial U_2}{\partial t} + 4\frac{\partial U_1}{\partial x} + \frac{\partial U_2}{\partial x} = 0,$$

for  $t \in [0, 1]$  and  $x \in [0, 1]$ .

The initial conditions are

$$U_1(x, 0) = \exp(x), \quad U_2(x, 0) = \sin(x),$$

and the Dirichlet boundary conditions for  $U_1$  at  $x = 0$  and  $U_2$  at  $x = 1$  are given by the exact solution:

$$U_1(x, t) = \frac{1}{2}\{\exp(x + t) + \exp(x - 3t)\} + \frac{1}{4}\{\sin(x - 3t) - \sin(x + t)\},$$

$$U_2(x, t) = \exp(x - 3t) - \exp(x + t) + \frac{1}{2}\{\sin(x + t) + \sin(x - 3t)\}.$$

## 9.1 Program Text

```
*      D03PEF Example Program Text
*      Mark 16 Release. NAG Copyright 1993.
*      .. Parameters ..
      INTEGER          NOUT
      PARAMETER        (NOUT=6)
      INTEGER          NPDE, NPTS, NLEFT, NEQN, LISAVE, NWKRES, LRSAVE
      PARAMETER        (NPDE=2, NPTS=41, NLEFT=1, NEQN=NPDE*NPTS,
+                      LISAVE=NEQN+24, NWKRES=NPDE*(NPTS+21+3*NPDE)
+                      +7*NPTS+4, LRSAVE=11*NEQN+(4*NPDE+NLEFT+2)
+                      *NEQN+50+NWKRES)
*      .. Local Scalars ..
      DOUBLE PRECISION ACC, TOUT, TS
      INTEGER          I, IFAIL, IND, IT, ITASK, ITRACE
*      .. Local Arrays ..
      DOUBLE PRECISION EU(NPDE,NPTS), RSAVE(LRSAVE), U(NPDE,NPTS),
+                      X(NPTS)
      INTEGER          ISAVE(LISAVE)
*      .. External Subroutines ..
      EXTERNAL          BNDARY, D03PEF, EXACT, PDEDEF, UINIT
*      .. Executable Statements ..
      WRITE (NOUT,*) 'D03PEF Example Program Results'
      ITRACE = 0
      ACC = 0.1D-5
*
*      Set spatial-mesh points
*
      DO 20 I = 1, NPTS
          X(I) = (I-1.0D0)/(NPTS-1.0D0)
20  CONTINUE
*
      IND = 0
      ITASK = 1
*
      CALL UINIT(NPDE,NPTS,X,U)
*
*      Loop over output value of t
      TS = 0.0D0
      TOUT = 0.0D0
      DO 40 IT = 1, 5
          TOUT = 0.2D0*IT
          IFAIL = 1
*
          CALL D03PEF(NPDE,TS,TOUT,PDEDEF,BNDARY,U,NPTS,X,NLEFT,ACC,
+                  RSAVE,LRSAVE,ISAVE,LISAVE,ITASK,ITRACE,IND,IFAIL)
*
          IF (IFAIL.NE.0) THEN
              WRITE (NOUT,99991) IFAIL
              GO TO 60
          ELSE IF (IT.EQ.1) THEN
              WRITE (NOUT,99997) ACC, NPTS
              WRITE (NOUT,99999) X(5), X(13), X(21), X(29), X(37)
          END IF
*
*      Check against the exact solution
*
      CALL EXACT(TOUT,NPDE,NPTS,X,EU)
*
      WRITE (NOUT,99998) TS
      WRITE (NOUT,99995) U(1,5), U(1,13), U(1,21), U(1,29), U(1,37)
      WRITE (NOUT,99994) EU(1,5), EU(1,13), EU(1,21), EU(1,29),
+      EU(1,37)
      WRITE (NOUT,99993) U(2,5), U(2,13), U(2,21), U(2,29), U(2,37)
```

```

      WRITE (NOUT,99992) EU(2,5), EU(2,13), EU(2,21), EU(2,29),
+      EU(2,37)
40 CONTINUE
      WRITE (NOUT,99996) ISAVE(1), ISAVE(2), ISAVE(3), ISAVE(5)
*
60 CONTINUE
*
99999 FORMAT (' X          ',5F10.4,/)
99998 FORMAT (' T = ',F5.2)
99997 FORMAT (// ' Accuracy requirement = ',E10.3, ' Number of points = ',
+      I3,/)
99996 FORMAT (' Number of integration steps in time = ',I6,/' Number o',
+      'f function evaluations = ',I6,/' Number of Jacobian eval',
+      'uations = ',I6,/' Number of iterations = ',I6)
99995 FORMAT (' Approx U1',5F10.4)
99994 FORMAT (' Exact U1',5F10.4)
99993 FORMAT (' Approx U2',5F10.4)
99992 FORMAT (' Exact U2',5F10.4,/)
99991 FORMAT (1X,/1X,' ** D03PEF returned with IFAIL = ',I5)
END
*
      SUBROUTINE UINIT(NPDE,NPTS,X,U)
*      Routine for PDE initial values
*      .. Scalar Arguments ..
      INTEGER          NPDE, NPTS
*      .. Array Arguments ..
      DOUBLE PRECISION U(NPDE,NPTS), X(NPTS)
*      .. Local Scalars ..
      INTEGER          I
*      .. Intrinsic Functions ..
      INTRINSIC        EXP, SIN
*      .. Executable Statements ..
      DO 20 I = 1, NPTS
          U(1,I) = EXP(X(I))
          U(2,I) = SIN(X(I))
20 CONTINUE
      RETURN
      END
*
      SUBROUTINE PDEDEF(NPDE,T,X,U,UDOT,DUDX,RES,IRES)
*      .. Scalar Arguments ..
      DOUBLE PRECISION T, X
      INTEGER          IRES, NPDE
*      .. Array Arguments ..
      DOUBLE PRECISION DUDX(NPDE), RES(NPDE), U(NPDE), UDOT(NPDE)
*      .. Executable Statements ..
      IF (IRES.EQ.-1) THEN
          RES(1) = UDOT(1)
          RES(2) = UDOT(2)
      ELSE
          RES(1) = UDOT(1) + DUDX(1) + DUDX(2)
          RES(2) = UDOT(2) + 4.0D0*DUDX(1) + DUDX(2)
      END IF
      RETURN
      END
*
      SUBROUTINE BNDARY(NPDE,T,IBND,NOBC,U,UDOT,RES,IRES)
*      .. Scalar Arguments ..
      DOUBLE PRECISION T
      INTEGER          IBND, IRES, NOBC, NPDE
*      .. Array Arguments ..
      DOUBLE PRECISION RES(NOBC), U(NPDE), UDOT(NPDE)
*      .. Intrinsic Functions ..
      INTRINSIC        EXP, SIN
*      .. Executable Statements ..
      IF (IBND.EQ.0) THEN
          IF (IRES.EQ.-1) THEN
              RES(1) = 0.0D0
          ELSE
              RES(1) = U(1) - 0.5D0*(EXP(T)+EXP(-3.0D0*T)) -
+              0.25D0*(SIN(-3.0D0*T)-SIN(T))

```

```

      END IF
    ELSE
      IF (IRES.EQ.-1) THEN
        RES(1) = 0.0D0
      ELSE
        RES(1) = U(2) - EXP(1.0D0-3.0D0*T) + EXP(1.0D0+T) -
+         0.5D0*(SIN(1.0D0-3.0D0*T)+SIN(1.0D0+T))
      END IF
    END IF
    RETURN
  END

*
  SUBROUTINE EXACT(T,NPDE,NPTS,X,U)
*   Exact solution (for comparison purposes)
*   .. Scalar Arguments ..
  DOUBLE PRECISION T
  INTEGER          NPDE, NPTS
*   .. Array Arguments ..
  DOUBLE PRECISION U(NPDE,NPTS), X(NPTS)
*   .. Local Scalars ..
  INTEGER          I
*   .. Intrinsic Functions ..
  INTRINSIC        EXP, SIN
*   .. Executable Statements ..
  DO 20 I = 1, NPTS
    U(1,I) = 0.5D0*(EXP(X(I)+T)+EXP(X(I)-3.0D0*T)) +
+      0.25D0*(SIN(X(I)-3.0D0*T)-SIN(X(I)+T))
    U(2,I) = EXP(X(I)-3.0D0*T) - EXP(X(I)+T) + 0.5D0*(SIN(X(I)
+      -3.0D0*T)+SIN(X(I)+T))
  20 CONTINUE
  RETURN
  END

```

## 9.2 Program Data

None.

## 9.3 Program Results

D03PEF Example Program Results

```

Accuracy requirement = 0.100E-05 Number of points = 41

X          0.1000    0.3000    0.5000    0.7000    0.9000

T = 0.20
Approx U1   0.7845    1.0010    1.2733    1.6115    2.0281
Exact  U1   0.7845    1.0010    1.2733    1.6115    2.0281
Approx U2  -0.8352   -0.8159   -0.8367   -0.9128   -1.0609
Exact  U2  -0.8353   -0.8160   -0.8367   -0.9129   -1.0609

T = 0.40
Approx U1   0.6481    0.8533    1.1212    1.4627    1.8903
Exact  U1   0.6481    0.8533    1.1212    1.4627    1.8903
Approx U2  -1.5216   -1.6767   -1.8934   -2.1917   -2.5944
Exact  U2  -1.5217   -1.6767   -1.8935   -2.1917   -2.5945

T = 0.60
Approx U1   0.6892    0.8961    1.1747    1.5374    1.9989
Exact  U1   0.6892    0.8962    1.1747    1.5374    1.9989
Approx U2  -2.0047   -2.3434   -2.7677   -3.3002   -3.9680
Exact  U2  -2.0048   -2.3436   -2.7678   -3.3003   -3.9680

T = 0.80
Approx U1   0.8977    1.1247    1.4320    1.8349    2.3514
Exact  U1   0.8977    1.1247    1.4320    1.8349    2.3512
Approx U2  -2.3403   -2.8675   -3.5110   -4.2960   -5.2536
Exact  U2  -2.3405   -2.8677   -3.5111   -4.2961   -5.2537


```

$T = 1.00$   
 Approx U1    1.2470    1.5206    1.8828    2.3528    2.9519  
 Exact U1    1.2470    1.5205    1.8829    2.3528    2.9518  
 Approx U2    -2.6229    -3.3338    -4.1998    -5.2505    -6.5218  
 Exact U2    -2.6232    -3.3340    -4.2001    -5.2507    -6.5219

Number of integration steps in time = 149  
 Number of function evaluations = 399  
 Number of Jacobian evaluations = 13  
 Number of iterations = 323

#### Example Program

Solution,  $U(1,x,t)$ , of First-order System using Keller, Box and BDF



Solution,  $U(2,x,t)$ , of First-order System using Keller, Box and BDF

